

異質商品

繁體中文版

基礎概念 (Foundations) • The Valuation Engineer

Bert Craytor • 2026 年 8 月 4 日

版本 1.0.0

DOI: [10.67330/s1scsv65](https://doi.org/10.67330/s1scsv65)

繁體中文譯本，由人工智慧輔助翻譯並經編輯部審校；英文版為權威版本。

形式定義。 異質商品 (heterogeneous good) 是指不可相互替換的商品：每一件商品都是買方分別賦予價值的特徵的獨特組合，而這些特徵本身並不單獨交易。

直觀闡釋。 同質商品 (homogeneous good) 是任何一件都可以的商品：一桶 WTI 原油、一蒲式耳二號黃玉米、電網上的一千瓦小時電力、某支股票的一股。這些商品背後的分級標準或規格的存在，正是為了使單件商品可以相互替換，從而使價格適用於類別而非特定的那一件。異質商品則是具體那一件很重要的商品——是這棟房子，而不是那棟房子；是這輛中古車，而不是旁邊那一輛。

在技術意義上，兩個特點使商品成為異質商品。第一，買方在意多個可區分的屬性，而這些屬性在市場上的各件商品之間存在差異。第二，這些屬性因商品的生產技術而被捆綁在一起，無法單獨購買。你不可能單獨購買 200 平方英尺的建築面積；你不可能脫離一筆土地而取得一個學區。特徵成捆而來，市場為整個組合定價，而不是為特徵本身定價。

房地產是異質商品的典型代表，並且異常鮮明地體現了這一屬性。區位特徵是完全不可複製的：任何兩筆土地都不可能佔據同一位置，因此每一筆房地產至少擁有一項屬性（其座標）將它與市場上的所有其他房地產區分開來。實體、法律與環境屬性又在此之上疊加了更多的變異維度，使得整體的特徵向量對每一筆房地產而言都是獨一無二的。

估價師在何處遇到它。 市場比較法 (sales comparison approach) 中的每一項調整，都是對估價對象與可比案例乃異質商品這一事實的隱含承認。如果房地產是同質的，成交價格本身就可以直接比較，根本不需要比較網格。網格是一種透過對組合構成差異作出調整，來調和不同組合價格的嘗試。值得注意的是網格不做什麼：它從不將成交價格分解為各特徵的價值貢獻。傳統網格沒有價值貢獻一欄——只有逐項直接估計的調整。

成本法透過從組成部分重建組合來因應異質性——基礎、構架、裝修、折舊——並對每一部分分別定價。收益法則透過將組合抽象為單一的現金流輪廓來因應異質性，由資本化率與收益率把組合層面的異質性吸收進一個源自市場的乘數。三種估價方法都是對「房地產是異質商品」這一根本事實的方法論回應。它們的區別在於如何因應這一事實，而不在於是否面對它。

估價師在非正式場合同樣會遇到異質性：每當他們思考可比案例的選擇、配對銷售分析 (paired-sales analysis) 或「同類相比」的比較時。尋找在儘可能多的屬性上相似的可比案例，這一直覺正是實務工作者對異質性的回應：將估價對象與可比案例之間的組合距離最小化，組合中未被觀測的成分對成交價格差異的貢獻便隨之減小。

為何關乎可辯護性。 房地產的異質性正是估價方法論必須作為一門學科存在的原因。如果房地產可以相互替換，市場價值 (market value) 便可從標價直接觀測，估價將淪為查表。正因為並非如此，價值必須從相關但不相同的房地產的價格中推斷得出，而每一步推斷都是一次需要說明理由的方法論選擇。

正式承認異質性，可以釐清 USPAP 及各準則條款所關注的可辯護性所依賴的若干實務問題。可比案例的選擇，是在特徵空間中挑選與估價對象足夠接近、以致從中還原的隱含價格可以推廣的樣本。一項調整是否合理，取決於其背後的隱含價格能否從市場證據中得到支持。各估價方法

之間的最終調和，反映的是估價師對哪種方法最可信地因應了這筆房地產的異質性的判斷。一旦「異質商品」被點明為深層條件，上述每一個可辯護性問題都會變得更加清晰。

估價實例演算。加利福尼亞州帕西菲卡 (Pacifica, California) 同一社區最近的兩筆成交：

屬性	可比案例 A	可比案例 B
建築面積 (平方英尺)	1,650	1,650
臥室 / 衛浴	3 / 2	3 / 2
地塊 (平方英尺)	9,000	7,200
景觀	有	無
成交價格	\$1,425,000	\$1,280,000

在同質商品的視角下，這是同一筆房地產：建築面積相同、房間數相同、社區相同。在異質商品的視角下，它們是兩個不同的組合，在買方分別賦予價值的兩項特徵（地塊與景觀）上存在差異。\$145,000 的價格差是這兩項屬性的聯合貢獻——兩個隱含價格糾纏在同一個配對之中。

要將隱含價格逐一還原，需要更多的配對，最好是每次只改變一項屬性。再有三四筆涵蓋有景觀 / 無景觀、大地塊 / 小地塊各種組合的成交，兩個隱含價格便開始可以分別識別。當足夠多的可比案例提供了足夠的變異時，特徵價格迴歸 (hedonic regression) 便可正式完成這一解纏；而當變異有限時，估價師只能侷限於區間化的配對銷售推理，其可靠性明確取決於各組合在特徵空間中彼此有多接近。

這個例子說明了為什麼異質性是根本條件：即使是兩棟「完全相同」的房子也並不相同，而估價的方法論工具之所以存在，正是為了處理它們之間的距離。

參考文獻。

Lancaster, K. J. (1966). A New Approach to Consumer Theory. *Journal of Political Economy*, 74(2), 132–157.

Rosen, S. (1974). Hedonic Prices and Implicit Markets: Product Differentiation in Pure Competition. *Journal of Political Economy*, 82(1), 34–55.

交叉引用：特徵空間 (*characteristics space*)；特徵價格函數 (*hedonic price function*)；隱含價格 (*implicit price*)。

附錄：Prolog 形式化

以下機器可讀的配套檔案 (001-heterogeneous-good.pl) 在本刊的 Prolog 知識庫中對本條目進行了形式化。該檔案同時作為可瀏覽的 Prolog 版本隨本文一併發布。

```
%% 001-heterogeneous-good.pl
%%
%% Prolog companion to Foundations Vol. 1, Issue 1, Article 001
%% Title: Heterogeneous Good
%% Author: Bert Craytor
%% Concept: heterogeneous_good
%% Version: 0.1.0-draft
%%
%% This file is the Prolog galley for the article: it encodes the
%% article's main claims, worked example, and inference rules in a form
%% that other Prolog code in the journal can consult, query, and verify
%% against. Read alongside the entry.tex / entry.pdf prose.
```

```
% =====
% DICTIONARY IMPORT
% =====
%
% The journal-wide dictionary is the source of truth for term definitions
% and relation schemas. The use_module below pulls those facts in at
% runtime; the comments that follow reproduce the relevant entries so a
% reader skimming this file in isolation can see what's been imported.

:- use_module('../../../tools/dictionary/dictionary').
:- dictionary_release('dictionary-2026.1').

% --- From the dictionary (loaded via :- use_module above) -----
% term(heterogeneous_good, "Heterogeneous Good", economic_theory,
%     "Heterogeneous goods are goods that are not interchangeable
%     because each is a distinct bundle of characteristics that buyers
%     value separately, and whose characteristics are not separately
%     traded.", published, 'dictionary-2026.1').
%
% term(characteristics_space, "Characteristics Space", economic_theory,
%     "The characteristics space of a class of heterogeneous goods is
%     the vector space whose coordinates index the attributes that
%     buyers value, and in which each good is represented by the
%     vector of its attribute levels.", published, 'dictionary-2026.1').
%
% term(implicit_price, "Implicit Price", economic_theory,
%     "The implicit price of a characteristic is the partial derivative
%     of the hedonic price function with respect to that characteristic,
%     evaluated at a point in the characteristics space; it represents
%     the marginal market price of an incremental unit of the
%     characteristic, holding all other characteristics fixed.",
%     published, 'dictionary-2026.1').
%
% term(hedonic_price_function, ...).
%
% relation(depends_on, 2, "depends on", [foundations_concept, foundations_concept],
%     foundations_methodology, "...", accepted, 'dictionary-2026.1').
% relation(related, 2, ...).
%
% These facts are NOT declared in this file; the use_module above is
% what makes them available at runtime.
% -----

:- use_terms([ % Foundations concepts (terms/foundations.pl)
    heterogeneous_good, characteristics_space, implicit_price,
    hedonic_price_function,
    % Practice terms (terms/valuation-practice.pl)
    real_estate, homogeneous_good,
    sales_comparison_approach, cost_approach, income_approach,
    comp_selection, paired_sales_analysis, adjustment,
    reconciliation, uspap, standards_rule ]).
:- use_relations([related/2, depends_on/2]).

% =====
% ARTICLE SUBJECT
% =====

article_concept(heterogeneous_good).
article_issue(volume(1), number(1), year(2026)).
```

```

article_id('foundations.2026.001').

% =====
% ARTICLE-LOCAL VOCABULARY
% =====
%
% Terms the article needs that are NOT yet in the dictionary, and that
% the editor decided are too narrow to promote. These are field names
% for the worked-example property records — not concepts in the
% journal's ontology.

local_term(gla_sqft).
local_term(bedrooms).
local_term(baths).
local_term(lot_sqft).
local_term(view).

% =====
% FORMAL DEFINITION (restated as Prolog rule)
% =====
%
% From the formal definition: a good is heterogeneous IF buyers care
% about multiple distinguishable attributes that vary across items in
% the market, AND those attributes are bundled by production rather
% than separately traded.

%! heterogeneous(?Good) is nondet.
%
% True when Good satisfies both bundling criteria. The two predicates
% buyers_value_multiple_attributes/1 and attributes_bundled_by_production/1
% must be asserted for specific goods elsewhere; here they characterize
% the formal definition itself.

heterogeneous(Good) :-
    buyers_value_multiple_attributes(Good),
    attributes_bundled_by_production(Good).

% =====
% INTUITIVE FRAMING — CONTRASTS WITH HOMOGENEOUS GOODS
% =====

% Canonical examples of homogeneous goods cited in the article.
canonical_homogeneous_example(wti_crude).
canonical_homogeneous_example(no_2_yellow_corn).
canonical_homogeneous_example(kilowatt_hour).
canonical_homogeneous_example(common_stock_share).

% Their distinguishing feature: grading or specification standards make
% individual items interchangeable.
interchangeable_via_grading(wti_crude).
interchangeable_via_grading(no_2_yellow_corn).
interchangeable_via_specification(kilowatt_hour).
interchangeable_via_specification(common_stock_share).

% Canonical heterogeneous goods.
canonical_heterogeneous_example(real_estate).
canonical_heterogeneous_example(used_car).
canonical_heterogeneous_example(unique_machinery).

```

```

% Why real estate exhibits heterogeneity unusually strongly:
% locational characteristics are perfectly non-replicable.
buyers_value_multiple_attributes(real_estate).
attributes_bundled_by_production(real_estate).
non_replicable_attribute(real_estate, location).

% =====
% APPRAISER ENCOUNTERS
% =====
%
% The three classical valuation approaches are all methodological
% responses to heterogeneity. They differ in HOW they navigate the
% bundle, not in WHETHER they confront it.

valuation_approach(sales_comparison_approach).
valuation_approach(cost_approach).
valuation_approach(income_approach).

responds_to_heterogeneity(sales_comparison_approach,
    "Reconciles the prices of distinct bundles by adjusting for composition
    ↳ differences; the grid records only adjustments, estimated directly, and never
    ↳ decomposes a price into feature value contributions.").
responds_to_heterogeneity(cost_approach,
    "Reconstructs the bundle from priced components: foundation, framing, finishes,
    ↳ depreciation.").
responds_to_heterogeneity(income_approach,
    "Abstracts the bundle into a single cash-flow profile; cap/yield rates absorb
    ↳ heterogeneity.").

% Practitioner-level encounters with heterogeneity.
practitioner_response_to_heterogeneity(comp_selection,
    "Minimize bundle distance between subject and comp in characteristics space.").
practitioner_response_to_heterogeneity(paired_sales_analysis,
    "Isolate an attribute's contribution by holding other attributes constant.").
practitioner_response_to_heterogeneity(adjustment,
    "Quantify the price impact of one differing attribute at a time.").

% =====
% DEFENSIBILITY
% =====
%
% Why naming "heterogeneous good" sharpens USPAP defensibility questions.

defensibility_consequence(comp_selection,
    "Comps must be close enough in characteristics space that recovered implicit
    ↳ prices generalize.").
defensibility_consequence(adjustment,
    "Each adjustment must be supportable from market evidence of the implicit price.")
↳ .
defensibility_consequence(reconciliation,
    "Choice among approaches reflects which best navigates THIS property's
    ↳ heterogeneity.").

% Root claim: valuation methodology exists as a discipline BECAUSE
% real estate is heterogeneous. If it were homogeneous, value would be
% directly observable from posted prices.
root_claim(
    "Valuation methodology exists as a discipline because real estate is heterogeneous
    ↳; in the homogeneous case, value would be directly observable.").

```

```
% =====
% WORKED EXAMPLE — PACIFICA COMPS
% =====
%
% Two recent sales in the same Pacifica, California neighborhood. The
% example demonstrates that even "identical" houses are not identical:
% the $145,000 price spread is the joint contribution of differences in
% lot size and view, two implicit prices entangled in a single pair.

% The 8-comp Pacifica dataset is shared across entries 001-006 and
% lives in kb/pacifica_comps.pl. Consulting it here brings property/1,
% in_market/2, attribute_value/3, and sale_price/2 into this article's
% namespace.
:- consult('../..kb/pacifica_comps').

% --- Derived facts from the example -----

% The attributes that differ between the two comps.
differing_attribute(comp_a, comp_b, lot_sqft).
differing_attribute(comp_a, comp_b, view).

% The price differential and what it's attributable to.
price_differential(comp_a, comp_b, 145000).
attributed_to(price_differential(comp_a, comp_b, 145000),
               [implicit_price_of(lot_sqft), implicit_price_of(view)]).

% --- Article's argument about identification -----

% A single pair cannot separately identify the two implicit prices —
% they are entangled in the single $145K observation.
joint_identification_only(price_differential(comp_a, comp_b, 145000),
                           [lot_sqft, view]).

% Separate identification requires more pairs varying one attribute at
% a time. With sufficient variation, hedonic regression performs the
% disentanglement formally; with limited variation, the appraiser uses
% bracketed paired-sales reasoning, whose reliability depends on how
% close the bundles are in characteristics space.
identification_strategy(few_pairs,
                        "Bracketed paired-sales reasoning. Reliability depends on bundle proximity in
                        ↪ characteristics space.").
identification_strategy(many_pairs_with_variation,
                        "Hedonic regression (see hedonic_regression)").

% =====
% INFERENCE RULES
% =====

%! comparable(?A, ?B) is nondet.
%
% A and B are comparable when they are distinct properties in the same
% market. This is the operational definition of "comp" used implicitly
% in the article.

comparable(A, B) :-
    property(A),
    property(B),
    A \== B,
```

```

    in_market(A, M),
    in_market(B, M).

%! attribute_difference(?A, ?B, ?Attr, ?DA, ?DB) is nondet.
%
% A and B differ on Attr, with DA the value at A and DB the value at B.
% Use this to enumerate the dimensions on which two comps actually vary.

attribute_difference(A, B, Attr, DA, DB) :-
    comparable(A, B),
    attribute_value(A, Attr, DA),
    attribute_value(B, Attr, DB),
    DA \== DB.

%! identifiable_implicit_price(?Attr) is nondet.
%
% An attribute's implicit price is identifiable from the observed comps
% if there exists at least one pair of comparables differing ONLY in
% that attribute. With our two-comp example this fails for lot_sqft AND
% view (both differ in the same pair), which is exactly the article's
% point.

identifiable_implicit_price(Attr) :-
    comparable(A, B),
    findall(X, attribute_difference(A, B, X, _, _), Diffs),
    Diffs == [Attr].

% =====
% CROSS-REFERENCES
% =====
%
% Concepts the article cross-references (corresponding to the
% \foundationsref{} macros in entry.tex).

cross_reference(characteristics_space).
cross_reference(hedonic_price_function).
cross_reference(implicit_price).

```